

Computação Algébrica no Cálculo das Variações: Determinação de Simetrias e Leis de Conservação

P.D.F. GOUVEIA¹, D.F.M. TORRES², Control Theory Group (*cotg*), Centro de Estudos em Optimização e Controlo (CEOC), Universidade de Aveiro, Departamento de Matemática, 3810-193 Aveiro, Portugal.

Resumo. Os problemas de otimização dinâmica (em espaços de funções) tratados pelo cálculo das variações são normalmente resolvidos por recurso às condições necessárias de Euler-Lagrange, que são equações diferenciais de segunda ordem (ou de ordem superior, quando os problemas variacionais envolvem derivadas de ordem superior a um). Estas equações são, em geral, não lineares e de difícil resolução. Uma forma de as simplificar consiste em obter leis de conservação: primeiros integrais das equações diferenciais de Euler-Lagrange. Se em áreas como a Física e a Economia a questão da existência de leis de conservação é resolvida de forma bastante natural, a própria aplicação sugerindo as leis de conservação (e.g. conservação de energia, conservação da quantidade de movimento, conservação do rendimento, etc.), de um ponto de vista estritamente matemático, dado um problema do cálculo de variações, o processo de obtenção das leis de conservação ou, até mesmo, a demonstração de que elas existem (ou não), deixa de ser uma questão óbvia. Neste trabalho mostramos como um sistema de computação algébrica como o Maple pode ser muito útil na abordagem a estas questões. Especificamente, propomos, como principal contribuição do nosso trabalho, um conjunto de facilidades computacionais simbólicas que permitem, de uma forma sistemática e automática, identificar as leis de conservação de uma dada funcional integral do cálculo das variações.

Mathematics Subject Classification 2000: 49-04, 49K05, 49S05.

1. Introdução

O cálculo das variações é uma área clássica da matemática, com mais de três séculos de idade, extremamente ativa no século XXI, e com inúmeras aplicações práticas na mecânica, economia, ciências dos materiais, ciências do espaço e engenharia. O cálculo das variações está na origem de muitas áreas mais recentes, como sejam a análise funcional e o controlo óptimo [4, 5, 6].

A computação algébrica, também chamada de computação científica ou simbólica, permite trabalhar com expressões matemáticas de maneira simbólica, não numérica, e é uma área de investigação moderna, que surgiu na segunda metade do século

¹pgouveia@ipb.pt; agradece a dispensa de serviço docente, para efeitos de doutoramento, concedida pelo Instituto Politécnico de Bragança ao abrigo do programa PRODEP III/5.3/2003.

²delfim@mat.ua.pt; sócio n^o89 da Associação Portuguesa de Controlo Automático (APCA).

XX. Se é verdade que os atuais sistemas de computação algébrica são extremamente poderosos, também não é menos verdade que muito existe por fazer no sentido de se colocarem os computadores a realizar tarefas matemáticas verdadeiramente interessantes.

O uso da teoria do cálculo das variações, na resolução de problemas concretos, requer cálculos não numéricos: no cálculo das variações a presença de fenômenos como o de Lavrentiev [3], faz com que as soluções algébricas exatas sejam muito mais convenientes do que as soluções numéricas. Tendo em conta que a resolução dos problemas do cálculo das variações passa quase obrigatoriamente pela resolução das equações diferenciais de Euler-Lagrange, equações estas, em geral, de difícil resolução, este trabalho tem como principal objectivo o desenvolvimento de um conjunto de procedimentos computacionais algébricos que permitem automatizar o processo de obtenção de leis de conservação (primeiros integrais das equações de Euler-Lagrange). Como é bem conhecido da teoria das equações diferenciais, estas leis de conservação são de extrema utilidade, permitindo baixar a ordem das equações [10].

2. Cálculo das Variações

Neste trabalho consideramos problemas do cálculo das variações de ordem superior: minimizar uma funcional integral

$$J[\mathbf{x}(\cdot)] = \int_a^b L(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \dots, \mathbf{x}^{(m)}(t)) dt, \quad (2.1)$$

onde o Lagrangeano L é uma função real que assumimos ser continuamente diferenciável em $[a, b] \times \mathbb{R}^{n \times (m+1)}$; $t \in \mathbb{R}$, é a variável independente; $\mathbf{x}(t) = [x_1(t) \ x_2(t) \ \dots \ x_n(t)]^T \in \mathbb{R}^n$, as variáveis dependentes; $\mathbf{x}^{(i)}(t) = [\frac{d^i x_1(t)}{dt^i} \ \frac{d^i x_2(t)}{dt^i} \ \dots \ \frac{d^i x_n(t)}{dt^i}]^T \in \mathbb{R}^n$, com $i = 1, \dots, m$, as derivadas de ordem i das variáveis dependentes em ordem a t ; e $\dot{\mathbf{x}}(t) \equiv \mathbf{x}^{(1)}(t)$. Assim, o Lagrangeano considerado depende explicitamente de uma variável independente, de n variáveis dependentes e das suas m primeiras derivadas. Para $m = 1$ obtemos o problema fundamental do cálculo das variações.

Na minimização da funcional (2.1) é usual recorrer-se ao sistema de equações de Euler-Lagrange

$$\frac{\partial L}{\partial \mathbf{x}} + \sum_{i=1}^m (-1)^i \frac{d^i}{dt^i} \left(\frac{\partial L}{\partial \mathbf{x}^{(i)}} \right) = \mathbf{0}, \quad (2.2)$$

onde $\frac{\partial L}{\partial \mathbf{x}^{(i)}} = [\frac{\partial L}{\partial x_1^{(i)}} \ \frac{\partial L}{\partial x_2^{(i)}} \ \dots \ \frac{\partial L}{\partial x_n^{(i)}}]$ e L , e suas derivadas, são avaliadas ao longo de $(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \dots, \mathbf{x}^{(m)}(t))$.

Definição 2.1. Às soluções das equações de Euler-Lagrange chamamos extremais.

Observação 2.1. Para o problema fundamental do cálculo das variações, i.e., quando a funcional não depende de derivadas de $\mathbf{x}(t)$ de maior ordem do que a primeira ($m = 1$), o sistema de equações de Euler-Lagrange (2.2) reduz-se a

$$\frac{\partial L}{\partial \mathbf{x}} - \frac{d}{dt} \left(\frac{\partial L}{\partial \dot{\mathbf{x}}} \right) = \mathbf{0}.$$

Observação 2.2. *Se a funcional não envolver mais do que uma variável dependente ($n = 1$), todas as grandezas presentes em (2.1) e (2.2) são escalares.*

Definição 2.2. *Uma função $t \rightarrow \phi(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \dots, \mathbf{x}^{(k)}(t))$, $k < 2m$, que se mantenha constante ao longo de todas as extremas do problema (2.1), é chamada de primeiro integral (de ordem k) da equação de Euler-Lagrange (2.2). À equação*

$$\phi(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \dots, \mathbf{x}^{(k)}(t)) = \text{const}$$

chamamos lei de conservação (de ordem k).

As leis de conservação são muito úteis, pois permitem reduzir a ordem das equações diferenciais de Euler-Lagrange. Com um número suficientemente elevado de primeiros integrais independentes, é mesmo possível determinar explicitamente as extremas.

Coloca-se, então, a seguinte questão: *Dada uma funcional integral do tipo (2.1), como obter leis de conservação?* Esta questão foi resolvida por Emmy Noether em 1918 [9]: se a funcional for invariante sob determinado tipo de transformações (transformações de invariância ou simetrias), então existem fórmulas explícitas para as leis de conservação. A dificuldade na sua aplicação reside na obtenção das simetrias (na obtenção das transformações de invariância). Neste trabalho automatizamos, por recurso ao sistema de computação algébrica Maple [1], a determinação de simetrias e a correspondente aplicação do Teorema de Noether.

3. Simetrias Variacionais

Para o estudo das propriedades de invariância das funcionais do cálculo das variações considera-se uma família uni-paramétrica de transformações $\mathbf{h}^s(t, \mathbf{x})$ que forma um grupo local de Lie [8, 2]. A família uni-paramétrica $\mathbf{h}^s(t, \mathbf{x})$ representa um conjunto de $n + 1$ transformações de $[a, b] \times \mathbb{R}^n$ em \mathbb{R}

$$t^s = h_t^s(t, \mathbf{x}), \quad x_i^s = h_{x_i}^s(t, \mathbf{x}), \quad \text{com } i = 1, \dots, n, \quad (3.1)$$

a que correspondem $n + 1$ geradores infinitesimais representados por

$$T(t, \mathbf{x}) = \left. \frac{\partial}{\partial s} h_t^s(t, \mathbf{x}) \right|_{s=0}, \quad X_i(t, \mathbf{x}) = \left. \frac{\partial}{\partial s} h_{x_i}^s(t, \mathbf{x}) \right|_{s=0}, \quad \text{com } i = 1, \dots, n. \quad (3.2)$$

Definição 3.1. *A funcional (2.1) diz-se invariante no intervalo $[a, b]$ sob as transformações uni-paramétricas (3.1) se, para todo o s suficientemente pequeno,*

$$\int_{\alpha}^{\beta} L(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \dots, \mathbf{x}^{(m)}(t)) dt = \int_{\alpha^s}^{\beta^s} L(t^s, \mathbf{x}^s(t^s), \dot{\mathbf{x}}^s(t^s), \dots, \mathbf{x}^{s(m)}(t^s)) dt^s$$

em qualquer subintervalo $[\alpha, \beta] \subseteq [a, b]$; com $\alpha^s = h_t^s(\alpha, \mathbf{x}(\alpha))$ e $\beta^s = h_t^s(\beta, \mathbf{x}(\beta))$.

Observação 3.1. *Nas condições da Definição 3.1 as transformações uni-paramétricas (3.1) constituem uma simetria variacional da funcional (2.1).*

O teorema que se segue estabelece uma condição necessária e suficiente de invariância, de extrema importância para os objectivos a que nos propomos.

Teorema 3.1 ([13]). *A funcional (2.1) é invariante sob as transformações uni-paramétricas (3.1), com geradores infinitesimais T e \mathbf{X} (3.2), se, e apenas se,*

$$\frac{\partial L}{\partial t}T + \sum_{i=0}^m \frac{\partial L}{\partial \mathbf{x}^{(i)}} \cdot \mathbf{p}^i + L \frac{dT}{dt} = 0, \quad (3.3)$$

onde

$$\mathbf{p}^0 = \mathbf{X}, \quad \mathbf{p}^{i+1} = \frac{d\mathbf{p}^i}{dt} - \mathbf{x}^{(i+1)} \frac{dT}{dt}, \quad i = 0, \dots, m-1. \quad (3.4)$$

Em (3.3) e (3.4) assumimos que T e $\mathbf{X} = [X_1 \ X_2 \ \dots \ X_n]^T$ são avaliadas em função de (t, \mathbf{x}) e \mathbf{p}^i em função de $(t, \mathbf{x}, \dot{\mathbf{x}}, \dots, \mathbf{x}^{(i)})$, com $i = 0, 1, \dots, m$.

Observação 3.2. *Todas as derivadas totais presentes em (3.3) e (3.4) podem ser expressas por derivadas parciais, usando as igualdades $\frac{dT}{dt} = \frac{\partial T}{\partial t} + \frac{\partial T}{\partial \mathbf{x}} \cdot \dot{\mathbf{x}}$ e $\frac{d\mathbf{p}^i}{dt} = \frac{\partial \mathbf{p}^i}{\partial t} + \sum_{k=0}^i \frac{\partial \mathbf{p}^i}{\partial \mathbf{x}^{(k)}} \cdot \mathbf{x}^{(k+1)}$, $i = 0, \dots, m-1$, onde*

$$\frac{\partial \mathbf{p}^i}{\partial \mathbf{x}^{(k)}} = \left[\frac{\partial \mathbf{p}^i}{\partial x_1^{(k)}} \quad \frac{\partial \mathbf{p}^i}{\partial x_2^{(k)}} \quad \dots \quad \frac{\partial \mathbf{p}^i}{\partial x_n^{(k)}} \right] = \begin{bmatrix} \frac{\partial p_1^i}{\partial x_1^{(k)}} & \frac{\partial p_1^i}{\partial x_2^{(k)}} & \dots & \frac{\partial p_1^i}{\partial x_n^{(k)}} \\ \frac{\partial p_2^i}{\partial x_1^{(k)}} & \frac{\partial p_2^i}{\partial x_2^{(k)}} & \dots & \frac{\partial p_2^i}{\partial x_n^{(k)}} \\ \vdots & \vdots & \ddots & \vdots \\ \frac{\partial p_n^i}{\partial x_1^{(k)}} & \frac{\partial p_n^i}{\partial x_2^{(k)}} & \dots & \frac{\partial p_n^i}{\partial x_n^{(k)}} \end{bmatrix}.$$

O Teorema 3.1, para além de servir de teste à existência de simetrias, estabelece um algoritmo para a determinação dos correspondentes geradores infinitesimais. Como veremos, este fato é crucial: o teorema de Noether (Teorema 4.1) afirma que as leis de conservação associadas a uma dada simetria variacional apenas dependem dos geradores infinitesimais.

Dado um Lagrangeano L , determinamos os geradores infinitesimais T e \mathbf{X} de uma família uni-paramétrica de transformações simétricas pelo seguinte método. A equação (3.3) é uma equação diferencial nas $n+1$ funções incógnitas T, X_1, X_2, \dots, X_n , que pretendemos determinar. Porém, a equação tem de permanecer válida para todos os $x_i, i = 1, \dots, n$. Como as funções T, X_1, X_2, \dots, X_n dependem de t e $x_i, i = 1, \dots, n$, ao substituirmos, na equação (3.3), L e todas as suas derivadas parciais pelos seus valores, obtemos um polinômio nas $n \times m$ variáveis $\dot{x}_1, \dots, \dot{x}_n, x_1^{(2)}, \dots, x_n^{(2)}, \dots, x_1^{(m)}, \dots, x_n^{(m)}$ e suas potências. Para que a equação seja válida para todos os valores das variáveis do polinômio, todos os seus coeficientes devem ser nulos. Notamos que os termos do polinômio poderão ser em maior número que as incógnitas do problema ($n+1$), pelo que a condição necessária e suficiente (3.3) pode conduzir a um sistema de equações sem solução. Tal fato significa apenas que nem todas as funcionais integrais do cálculo das variações admitem simetrias variacionais. O sistema de equações a resolver, para a obtenção dos geradores, é um sistema de equações diferenciais às derivadas parciais. No entanto, ao contrário

das equações diferenciais ordinárias de Euler-Lagrange, em geral não lineares e de difícil resolução, este sistema é linear em $\frac{\partial T}{\partial t}$, $\frac{\partial T}{\partial \mathbf{x}}$, $\frac{\partial \mathbf{X}}{\partial t}$ e $\frac{\partial \mathbf{X}}{\partial \mathbf{x}}$.

A resolução do sistema de equações diferenciais parciais que deriva da expressão (3.3), nomeadamente quando lidamos com valores de n e m superiores à unidade, envolve um número muito elevado de cálculos, o que torna premente a necessidade de nos munirmos de ferramentas computacionais que automatizem o trabalho. Com esse fim, desenvolvemos um procedimento em `Maple`, designado *Simetria* (ver §6).

Na seção que se segue mostramos como os geradores, obtidos por intermédio do nosso procedimento *Simetria*, podem ser usados na obtenção explícita de leis de conservação.

4. Leis de Conservação

Emmy Noether foi a primeira a estabelecer a relação entre a existência de simetrias e a existência de leis de conservação. Esta ligação constitui um princípio universal, passível de ser formulado na forma de teorema nos mais diversos contextos e sob as mais variadas hipóteses [7, 10, 11, 12, 13].

Teorema 4.1 (Teorema de Noether [13]). *Se a funcional (2.1) é invariante sob as transformações uni-paramétricas (3.1), com geradores infinitesimais T e \mathbf{X} , então*

$$\sum_{i=1}^m \Psi^i \cdot \mathbf{p}^{i-1} + \left(L - \sum_{i=1}^m \Psi^i \cdot \mathbf{x}^{(i)} \right) T = \text{const}, \quad t \in [a, b], \quad (4.1)$$

com

$$\Psi^m = \frac{\partial L}{\partial \mathbf{x}^{(m)}}, \quad \Psi^{i-1} = \frac{\partial L}{\partial \mathbf{x}^{(i-1)}} - \frac{d\Psi^i}{dt}, \quad i = m, m-1, \dots, 2,$$

$$\frac{d\Psi^i}{dt} = \frac{\partial \Psi^i}{\partial t} + \sum_{k=0}^{2m-i} \left(\mathbf{x}^{(k+1)} \right)^T \cdot \frac{\partial \Psi^i}{\partial \mathbf{x}^{(k)}},$$

$$\text{onde } \frac{\partial \Psi^i}{\partial \mathbf{x}^{(k)}} = \begin{bmatrix} \frac{\partial \Psi^i}{\partial x_1^{(k)}} \\ \frac{\partial \Psi^i}{\partial x_2^{(k)}} \\ \vdots \\ \frac{\partial \Psi^i}{\partial x_n^{(k)}} \end{bmatrix} = \begin{bmatrix} \frac{\partial \psi_1^i}{\partial x_1^{(k)}} & \frac{\partial \psi_2^i}{\partial x_1^{(k)}} & \cdots & \frac{\partial \psi_n^i}{\partial x_1^{(k)}} \\ \frac{\partial \psi_1^i}{\partial x_2^{(k)}} & \frac{\partial \psi_2^i}{\partial x_2^{(k)}} & \cdots & \frac{\partial \psi_n^i}{\partial x_2^{(k)}} \\ \vdots & \vdots & \ddots & \vdots \\ \frac{\partial \psi_1^i}{\partial x_n^{(k)}} & \frac{\partial \psi_2^i}{\partial x_n^{(k)}} & \cdots & \frac{\partial \psi_n^i}{\partial x_n^{(k)}} \end{bmatrix}$$

considerando a grandeza Ψ^i avaliada em $(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \dots, \mathbf{x}^{(2m-i)}(t))$, $i = 1, \dots, m$.

As leis de conservação que procuramos são obtidas substituindo nas equações (3.4) e (4.1), os geradores infinitesimais T e \mathbf{X} encontrados pelo método descrito na seção anterior. Na Seção 6. definimos o procedimento *Noether*. Este procedimento tem por entradas os geradores infinitesimais, que são obtidos por intermédio do nosso procedimento *Simetria*, e como saída a correspondente lei de conservação (4.1). Resumindo: dado um problema do cálculo das variações (2.1), obtemos as leis de conservação, de uma forma automática, através de um processo de duas etapas:

com o nosso procedimento *Simetria* obtemos todas as possíveis simetrias do problema; recorrendo depois ao nosso procedimento *Noether*, baseado no Teorema 4.1, obtemos as correspondentes leis de conservação. Na seção seguinte apresentamos um exemplo que ilustra todo o processo.

5. Exemplo Ilustrativo

Consideramos agora uma situação concreta, mostrando a funcionalidade e a utilidade das ferramentas desenvolvidas. Vejamos então um caso relativamente simples de um Lagrangeano com duas variáveis dependentes ($n = 2$) e duas derivadas de ordem superior ($m = 2$): $L(t, \mathbf{x}, \dot{\mathbf{x}}, \ddot{\mathbf{x}}) = \dot{x}_1^2 + \ddot{x}_2^2$.

Definindo $L = v_1^2 + a_2^2$ com a instrução **Maple**

```
> L:=v[1]^2+a[2]^2:
```

o nosso procedimento *Simetria* determina os geradores infinitesimais das simetrias do problema do cálculo das variações em consideração:

```
> Simetria(L, t, [x[1],x[2]], [v[1],v[2]], [a[1],a[2]]);
```

$$\{ T(t, x_1, x_2) = -C1 t + -C2,$$

$$X_1(t, x_1, x_2) = 1/2 \cdot C1 x_1 + -C5, X_2(t, x_1, x_2) = 3/2 \cdot C1 x_2 + -C3 t + -C4 \}$$

A família de geradores depende de cinco parâmetros que advêm das constantes de integração. A lei de conservação correspondente a estes geradores é facilmente obtida por intermédio do nosso procedimento *Noether*³:

```
> LC := Noether(L, t, [x[1],x[2]], [v[1],v[2]], [a[1],a[2]], %);
```

$$LC := 2(1/2 \cdot C1 x_1(t) + -C5) \frac{d}{dt} x_1(t) - 2(3/2 \cdot C1 x_2(t) + -C3 t + -C4) \frac{d^3}{dt^3} x_2(t)$$

$$+ 2 \left(-C3 + 1/2 \cdot C1 \frac{d}{dt} x_2(t) \right) \frac{d^2}{dt^2} x_2(t)$$

$$+ \left(- \left(\frac{d}{dt} x_1(t) \right)^2 - \left(\frac{d^2}{dt^2} x_2(t) \right)^2 + 2 \frac{d}{dt} x_2(t) \frac{d^3}{dt^3} x_2(t) \right) (-C1 t + -C2) = const$$

É, neste caso, muito fácil verificar a validade da lei de conservação obtida. Por definição, basta mostrar que a igualdade é verificada ao longo das extremais. A equação de Euler-Lagrange é um sistema de equações diferenciais de 4ª ordem⁴

```
> EulerLagrange(L, t, [x[1],x[2]], [v[1],v[2]], [a[1],a[2]]);
```

$$\left\{ -2 \frac{d^2}{dt^2} x_1(t) = 0, 2 \frac{d^4}{dt^4} x_2(t) = 0 \right\}$$

e as extremais são as suas soluções:

```
> dsolve(%);
```

$$\{ x_1(t) = -C6 t + -C7, x_2(t) = 1/6 \cdot C8 t^3 + 1/2 \cdot C9 t^2 + -C10 t + -C11 \}$$

Substituindo as extremais na lei de conservação, obtemos, como esperado, uma proposição verdadeira:

```
> expand(subs(% , LC));
```

$$-C6 \cdot C1 \cdot C7 + 2 \cdot C6 \cdot C5 - 3 \cdot C8 \cdot C1 \cdot C11 - 2 \cdot C8 \cdot C4 + 2 \cdot C3 \cdot C9$$

$$+ -C1 \cdot C10 \cdot C9 - C6^2 \cdot C2 - C9^2 \cdot C2 + 2 \cdot C8 \cdot C10 \cdot C2 = const$$

³O sinal de percentagem (%) é um operador usado em **Maple** para referenciar o resultado do comando anterior. Para uma introdução ao **Maple**, veja-se [1].

⁴Embora o **Maple 9**, através do seu *package VariationalCalculus*, disponha já de um procedimento para a definição da equação de Euler-Lagrange, optou-se por implementar um outro procedimento denominado *EulerLagrange* por uma questão de uniformização da forma de invocação.

6. Procedimentos Maple⁵

Os procedimentos *Simetria* e *Noether*, descritos e ilustrados nas seções anteriores, são agora definidos usando o sistema de computação algébrica Maple (versão 9).

Simetria determina as simetrias de um Lagrangeano de várias variáveis dependentes e com derivadas de ordem superior, de acordo com a Seção 3.

Devolve: conjunto/vetor de geradores infinitesimais das transformações simétricas.

Forma de invocação: `Simetria(L, t, x, x1, x2, ..., xm)`

Parâmetros:

- L - expressão do Lagrangeano;
- t - nome da variável independente;
- x - nome, lista de nomes ou vetor de nomes das variáveis dependentes;
- xi - nome, lista de nomes ou vetor de nomes das derivadas de ordem i das variáveis dependentes;

```

Simetria:=proc(L::algebraic,t::name,x0::{name,list(name),'Vector[column]'}
              (name)},x1::{name,list(name),'Vector[column]'}(name)})
local n,m,xx,P,EqD,SysEqD,Sol,xi,Tdt,soma,V,r,i,j,k;
if nargs<4 then print('Nº de args insuficiente.');
```

```

return;
elif not type([args[3..-1]],{'list'(name),'listlist'(name),
                          'list'('Vector[column]')(name)})
then print('Erro na lista das var. depend. ou suas derivadas.');
```

```

return;
end if;
unassign('T'); unassign('X');
xx:=convert(x0,'list')[]; n:=nops([xx]); m:=nargs-3;
xi:=[seq(Vector(convert(args[i],'list')),i=3..m+3)];
Tdt:=diff(T(t,xx),t)+Vector[row]([seq(diff(T(t,xx),i),i=xx)].xi[2]);
if n>1 then P:=[Vector([seq(X[i](t,xx),i=1..n)])];
else P:=[Vector([X(t,xx)])]; end if;
for i from 1 to m do
  V:=Vector(n);
  for k from 0 to i-1 do
    V:=V+Matrix(n,(r,j)->diff(P[i][r],xi[k+1][j])).xi[k+2];
  end do;
  P:=[P[],map(diff,P[i],t)+V-xi[i+1]*Tdt];
end do;
soma:=0;
for i from 0 to m do
  soma:=soma+Vector[row]([seq(diff(L,r),r=convert(xi[i+1],list))]).P[i+1];
end do;
EqD:=diff(L,t)*T(t,xx)+soma+L*Tdt;
EqD:=collect(EqD,[seq(convert(xi[i+1],'list')[],i=1..m)],distributed);

```

⁵Nota dos autores: publicam-se as definições das rotinas desenvolvidas por se considerar que a relevância do trabalho reside essencialmente na originalidade do código desenvolvido: o artigo é, na opinião dos autores, interessante, não propriamente por causa de uma nova teoria apresentada, mas pelas ferramentas computacionais desenvolvidas, que permitem a determinação automática de simetrias e leis de conservação no contexto do cálculo variacional, tornando deste modo úteis, na prática, resultados clássicos e recentes.

```

SysEqD:={coeffs(EqD,[seq(convert(xi[i+1], 'list') [], i=1..m)]);
Sol:=pdsolve(SysEqD,{T(t,xx)} union convert(P[1], 'set'));
if type(x0, 'Vector') then
  return (subs(Sol,T(t,xx)), Vector(subs(Sol,P[1])));
else return Sol; end if;
end proc:

```

Noether dados os geradores infinitesimais, determina a Lei de Conservação de um Lagrangeano de várias variáveis dependentes e com derivadas de ordem superior, de acordo com a Seção 4.

Devolve: lei de conservação.

Formas de invocação:

- Noether(L, t, x, x1, x2, ..., xm, S)
- Noether(L, t, xs, x1s, x2s, ..., xms, T, X)

Parâmetros:

- S - conjunto de geradores infinitesimais das simetrias (output do procedimento *Simetria*);
- xs - vetor de nomes das variáveis dependentes;
- xis - vetor de nomes das derivadas de ordem i das variáveis dependentes;
- T - gerador da transformação para a variável independente (t);
- X - vetor com os geradores das transformações para as variáveis dependentes (xs).

```

Noether:=proc(L::algebraic,t::name,x0::{name,list(name),'Vector[column]'}
              (name),x1::{name,list(name),'Vector[column]'}(name))
local xx,n,m,P,psi,LC,xi,Tdt,V,r,i,j,k;
if type(x0,'Vector') then m:=nargs-5; else m:=nargs-4; end if;
if m<1 then print('Nº de args insuficiente. '); return;
elif not type([args[3..3+m]],{'list'(name),'listlist'(name),
                              'list'('Vector[column]')(name)}) then
  print('Erro na lista das var. depend. ou suas derivadas. '); return;
elif (type(x0,'Vector') and not(type(args[-1],'Vector[column]') and
                                type(args[-2],algebraic))) or (not type(x0,'Vector') and
                                                                not type(args[-1],'set')) then
  print('Conj. de gerad. inv{\a}lido. '); return;
end if;
xx:=convert(x0,'list') []; n:=nops([xx]); unassign('T'); unassign('X');
xi:=seq(Vector(convert(args[i], 'list')), i=3..m+3);
xi:=xi[1], seq(Vector([seq(x||i[k], k=1..n)], i=m+1..2*m-1));
Tdt:=diff(T(t,xx), t)+Vector[row]([seq(diff(T(t,xx), i), i=xx)].xi[2]);
if n>1 then P:=Vector([seq(X[i](t,xx), i=1..n)]);
else P:=Vector([X(t,xx)]); end if;
for i from 1 to (m-1) do
  V:=Vector(n);
  for k from 0 to i-1 do
    V:=V+Matrix(n, (r, j)->diff(P[i][r], xi[k+1][j])).xi[k+2] end do;
  P:=P[1], map(diff, P[i], t)+V-xi[i+1]*Tdt;
end do;
psi:=Vector[row]([seq(diff(L, i), i=convert(xi[m+1], 'list'))]);
for i from m by -1 to 2 do

```



```

V:=Vector[row](n):
for k from 0 to 2*m-i do
  V:=V+LinearAlgebra[Transpose](xi[k+2]).Matrix(n,(j,r)->
    diff(psi[1][r],xi[k+1][j])); end do:
psi:=[Vector[row]([seq(diff(L,i),i=convert(xi[i],'list'))])
  -map(diff,psi[1],t)-V,psi[]]:
end do:
LC:=sum('psi[i].P[i]', 'i'=1..m)+(L-sum('psi[i].xi[i+1]', 'i'=1..m))*T(t,xx)
  =const:
if type(x0,'Vector') then
  LC:=eval(LC,[T(t,xx)=args[-2],seq(P[1][i]=args[-1][i],i=1..n)]);
else LC:=eval(LC,args[-1]); end if;
LC:=subs({map(i->i=i(t),[xx])[]},LC);
LC:=subs({seq(seq(xi[k+1][i]=diff(xi[1][i](t),t$k),i=1..n),k=1..2*m-1)},
  LC);

return LC;
end proc:

```

7. Conclusão e Trabalho Futuro

Os sistemas atuais de computação algébrica colocam à nossa disposição ambientes de computação científica extremamente sofisticados e poderosos. Proporcionam já muito conhecimento matemático e permitem estender esse conhecimento por intermédio de linguagens de programação de muito alto nível, expressivas e intuitivas, próximas da linguagem matemática. Neste trabalho disponibilizamos o código de novos procedimentos computacionais algébricos, desenvolvidos pelos autores usando o sistema de computação matemática **Maple 9**, que permitem a determinação automática de simetrias e leis de conservação no cálculo das variações.

Como trabalho futuro pretendemos estender o nosso “package Maple” ao caso discreto no tempo e aos problemas mais genéricos tratados pelo *Controlo Ótimo*.

Abstract. The problems of the calculus of variations are usually solved with the help of the Euler-Lagrange differential equations. These equations are, in general, nonlinear, and hard to solve. One way to address the problem is to obtain conservation laws. We develop symbolic computational facilities, based on a systematic method, which permits to compute conservation laws for the problems of the calculus of variations. The central result used is the celebrated Noether’s theorem, which connects the existence of conservation laws with the existence of invariance properties for the problem (with the existence of variational symmetries). We show how a modern Computer Algebra System can be used to find all the symmetries of the problems of the calculus of variations and the corresponding conservation laws.

Referências

- [1] L.N. de Andrade, “Introdução à Computação Algébrica com o Maple”, IMPA, editora da Sociedade Brasileira de Matemática, 2004.
- [2] B. van Brunt, “The Calculus of Variations”, Springer-Verlag New York, 2004.

- [3] G. Buttazzo e M. Belloni, A Survey on Old and Recent Results About the Gap Phenomenon in the Calculus of Variations, “Recent developments in well-posed variational problems”, pp. 1–27, Kluwer Acad. Publ., 1995.
- [4] F.S. David, “O Cálculo Variacional Clássico e Algumas das suas Aplicações à Física Matemática – Referência a Algumas Extensões mais Recentes”, Electricidade de Portugal, 1986.
- [5] L.P. Lebedev e M.J. Cloud, “The Calculus of Variations and Functional Analysis – with Optimal Control and Applications in Mechanics”, World Scientific, 2003.
- [6] A. Leitão, “Cálculo Variacional e Controle Ótimo”, IMPA, 23^o Colóquio Brasileiro de Matemática, 2001.
- [7] J.D. Logan, “Invariant Variational Principles”, Academic Press [Harcourt Brace Jovanovich Publishers], 1977.
- [8] J.D. Logan, “Applied Mathematics – A Contemporary Approach”, John Wiley & Sons, New York, 1987.
- [9] E. Noether, Invariante Variationsprobleme, *Gött. Nachr.* (1918), 235–257.
- [10] P.J. Olver, “Applications of Lie Groups to Differential Equations”, Springer-Verlag, 1986.
- [11] D.F.M. Torres, On the Noether Theorem for Optimal Control, *European Journal of Control*, **8**, No. 1 (2002), 56–63.
- [12] D.F.M. Torres, Integrals of Motion for Discrete-Time Optimal Control Problems, *Control Applications of Optimisations 2003*, R. Bars and E. Gyurkovics eds., IFAC Workshop Series (2003), 33–38.
- [13] D.F.M. Torres, Proper Extensions of Noether’s Symmetry Theorem for Nonsmooth Extremals of the Calculus of Variations, *Communications on Pure and Applied Analysis*, **3**, No. 3 (2004), 491–500.